Commutation relation in angular momentum

Masatsugu Sei Suzuki
Department of Physics, SUNY at Binghamton
(Date: November 13, 2013)

Position and momentum operators for a given axis do not commute, whereas position and momentum operators for different axes do commute,

$$[\hat{x}, \hat{p}_x] = i\hbar\hat{1}, \qquad [\hat{y}, \hat{p}_y] = i\hbar\hat{1}, \qquad [\hat{z}, \hat{p}_z] = i\hbar\hat{1},$$

and

$$[\hat{x}, \hat{p}_y] = [\hat{x}, \hat{p}_z] = 0, \quad [\hat{y}, \hat{p}_x] = [\hat{y}, \hat{p}_z] = 0, \quad [\hat{z}, \hat{p}_x] = [\hat{z}, \hat{p}_y] = 0,$$

$$[\hat{z}, \hat{x}] = [\hat{z}, \hat{y}] = [\hat{x}, \hat{y}].$$

Using these commutators, we calculate the commutators of the component of the angular momentum operator.

1. Definition of orbital angular momentum

The angular momentum is defined as

$$\hat{\boldsymbol{L}} = \hat{\boldsymbol{r}} \times \hat{\boldsymbol{p}} = \begin{vmatrix} \boldsymbol{e}_x & \boldsymbol{e}_y & \boldsymbol{e}_z \\ \hat{x} & \hat{y} & \hat{z} \\ \hat{p}_x & \hat{p}_y & \hat{p}_z \end{vmatrix},$$

$$\hat{L}_x = \hat{y}\hat{p}_z - \hat{z}\hat{p}_y,$$

$$\hat{L}_y = \hat{z}\hat{p}_x - \hat{x}\hat{p}_z,$$

$$\hat{L}_z = \hat{x}\hat{p}_y - \hat{y}\hat{p}_x.$$

2. Commutation relations

$$[\hat{L}_x, \hat{L}_y] = i\hbar\hat{L}_z,$$

or

$$
\begin{aligned}
[\hat{L}_x, \hat{L}_y] &= [\hat{y}\hat{p}_z - \hat{z}\hat{p}_y, \hat{z}\hat{p}_x - \hat{x}\hat{p}_z] \\
&= [\hat{y}\hat{p}_z, \hat{z}\hat{p}_x] + [\hat{z}\hat{p}_y, \hat{x}\hat{p}_z] \\
&= \hat{y}[\hat{p}_z, \hat{z}]\hat{p}_x + \hat{p}_y[\hat{z}, \hat{p}_z]\hat{x} \\
&= -\frac{\hbar}{i}(-\hat{y}\hat{p}_x + \hat{x}\hat{p}_y) = i\hbar\hat{L}_z
\end{aligned}
$$

or

$$[\hat{L}_x, \hat{L}_y] = i\hbar\hat{L}_z,$$

Similarly,

$$[\hat{L}_y, \hat{L}_z] = i\hbar\hat{L}_x, \qquad [\hat{L}_z, \hat{L}_x] = i\hbar\hat{L}_y.$$

$\hat{L}^2$ is defined by

$$\hat{L}^2 = \hat{L}_x{}^2 + \hat{L}_y{}^2 + \hat{L}_z{}^2.$$

We have

$$
\begin{aligned}
[\hat{L}^2, \hat{L}_z] &= [\hat{L}_x{}^2 + \hat{L}_y{}^2 + \hat{L}_z{}^2, \hat{L}_z] = [\hat{L}_x{}^2, \hat{L}_z] + [\hat{L}_y{}^2, \hat{L}_z] \\
&= -[\hat{L}_z, \hat{L}_x{}^2] - [\hat{L}_z, \hat{L}_y{}^2] = \hat{0}
\end{aligned}
$$

using the relations

$$[\hat{L}_z, \hat{L}_x{}^2] = [\hat{L}_z, \hat{L}_x]\hat{L}_x + \hat{L}_x[\hat{L}_z, \hat{L}_x] = i\hbar(\hat{L}_y\hat{L}_x + \hat{L}_x\hat{L}_y),$$

$$[\hat{L}_z, \hat{L}_y{}^2] = [\hat{L}_z, \hat{L}_y]\hat{L}_y + \hat{L}_y[\hat{L}_z, \hat{L}_y] = -i\hbar(\hat{L}_y\hat{L}_x + \hat{L}_x\hat{L}_y).$$

Similarly

$$[\hat{L}^2, \hat{L}_x] = \hat{0}, \quad [\hat{L}^2, \hat{L}_y] = \hat{0} \quad [\hat{L}^2, \hat{L}_z] = \hat{0}.$$

In general, the same relations hold for the general angular momentum ***J***.

$$[\hat{J}^2, \hat{J}_x] = \hat{0}, \quad [\hat{J}^2, \hat{J}_y] = \hat{0} \quad [\hat{J}^2, \hat{J}_z] = \hat{0},$$

with

$$\hat{J}^2 = \hat{J}_x{}^2 + \hat{J}_y{}^2 + \hat{J}_z{}^2,$$

$$[\hat{J}_x, \hat{J}_y] = i\hbar\hat{J}_z , \quad [\hat{J}_y, \hat{J}_z] = i\hbar\hat{J}_x , \quad [\hat{J}_z, \hat{J}_x] = i\hbar\hat{J}_y .$$

We define new operators by

$$\hat{J}_\pm = \hat{J}_x \pm i\hat{J}_y ,$$

$$[\hat{J}_z, \hat{J}_+] = \hbar\hat{J}_+ ,$$

$$[\hat{J}_z, \hat{J}_-] = -\hbar\hat{J}_- ,$$

$$[\hat{J}_+, \hat{J}_-] = 2\hbar\hat{J}_z ,$$

$$\hat{J}^2 - \hat{J}_z{}^2 + \hbar\hat{J}_z - \hat{J}_+\hat{J}_- = \hat{0} ,$$

$$\hat{J}^2 - \hat{J}_z{}^2 - \hbar\hat{J}_z - \hat{J}_-\hat{J}_+ = \hat{0} ,$$

$$\hat{J}^2 - \hat{J}_z{}^2 = \frac{1}{2}(\hat{J}_+\hat{J}_- + \hat{J}_-\hat{J}_+) .$$

2. **Mathematica**

Using Mathematica, we give the proof for various kinds of commutation relations for the angular momentum. We note the following.

$$\begin{aligned}\langle \boldsymbol{r} | \hat{L}_x | \psi \rangle &= \langle \boldsymbol{r} | \hat{y}\hat{p}_z - \hat{z}\hat{p}_y | \psi \rangle \\ &= \frac{\hbar}{i}(y\frac{\partial}{\partial z} - z\frac{\partial}{\partial y})\langle \boldsymbol{r} | \psi \rangle \\ &= \frac{\hbar}{i}(y\frac{\partial}{\partial z} - z\frac{\partial}{\partial y})\psi(\boldsymbol{r})\end{aligned}$$

$$\begin{aligned}\langle \boldsymbol{r} | \hat{L}_y | \psi \rangle &= \langle \boldsymbol{r} | \hat{z}\hat{p}_x - \hat{x}\hat{p}_z | \psi \rangle \\ &= \frac{\hbar}{i}(z\frac{\partial}{\partial x} - x\frac{\partial}{\partial z})\psi(\boldsymbol{r})\end{aligned}$$

$$\begin{aligned}\langle \boldsymbol{r} | \hat{L}_z | \psi \rangle &= \langle \boldsymbol{r} | \hat{x}\hat{p}_y - y\hat{p}_x | \psi \rangle \\ &= \frac{\hbar}{i}(x\frac{\partial}{\partial y} - y\frac{\partial}{\partial x})\psi(\boldsymbol{r})\end{aligned}$$

$$\langle \boldsymbol{r}|\hat{L}_x\hat{L}_y|\psi\rangle = \left(\frac{\hbar}{i}\right)^2 (y\frac{\partial}{\partial z} - z\frac{\partial}{\partial y})(z\frac{\partial}{\partial x} - x\frac{\partial}{\partial z})\psi(\boldsymbol{r}) ,$$

$$\langle \boldsymbol{r}|\hat{L}_x\hat{L}_x|\psi\rangle = \left(\frac{\hbar}{i}\right)^2 (y\frac{\partial}{\partial z} - z\frac{\partial}{\partial y})^2\psi(\boldsymbol{r}) ,$$

$$\hat{L}^2 = \hat{L}_x{}^2 + \hat{L}_y{}^2 + \hat{L}_z{}^2 , \qquad \hat{L}_\pm = \hat{L}_x \pm i\hat{L}_y .$$

Eq1: $[\hat{L}_x, \hat{L}_y] = i\hbar\hat{L}_z$,

Eq.2: $[\hat{L}_y, \hat{L}_z] = i\hbar\hat{L}_x$,

Eq.3: $[\hat{L}_z, \hat{L}_x] = i\hbar\hat{L}_y$

Eq.4: $[\hat{L}_z, \hat{L}_x{}^2] = i\hbar(\hat{L}_y\hat{L}_x + \hat{L}_x\hat{L}_y)$

Eq.5: $[\hat{L}_z, \hat{L}_y{}^2] = -i\hbar(\hat{L}_y\hat{L}_x + \hat{L}_x\hat{L}_y)$

Eq.6 $[\hat{L}_z, \hat{L}_z{}^2] = 0$

Eq.7 $[\hat{L}^2, \hat{L}_z] = \hat{0}$,

Eq.8 $[\hat{L}_z, \hat{L}_+] = \hat{0}$

Eq.9 $[\hat{L}_z, \hat{L}_-] = \hat{0}$

Eq.10 $\hat{\boldsymbol{L}}^2 - \hat{L}_z{}^2 + \hbar\hat{L}_z - \hat{L}_+\hat{L}_- = \hat{0}$

Eq.11 $\hat{\boldsymbol{L}}^2 - \hat{L}_z{}^2 - \hbar\hat{L}_z - \hat{L}_-\hat{L}_+ = \hat{0}$

Eq.12 $\hat{\boldsymbol{L}}^2 - \hat{L}_z{}^2 = \frac{1}{2}(\hat{L}_+\hat{L}_- + \hat{L}_-\hat{L}_+)$

Eq.13 Expression of $\langle \boldsymbol{r}|\hat{\boldsymbol{L}}^2|\psi\rangle$

((**Mathematica**))

```
Clear["Global`"];
Lz := ℏ/ⅈ (x D[#, y] - y D[#, x]) &;
Lx := ℏ/ⅈ (y D[#, z] - z D[#, y]) &;
Ly := ℏ/ⅈ (z  D[#, x] - x D[#, z]) &;
LLx := Nest[Lx, #, 2] &; LLy := Nest[Ly, #, 2] &;
LLz := Nest[Lz, #, 2] &;
LL := (LLx[#] + LLy[#] + LLz[#]) &;
Lp := (Lx[#] + ⅈ Ly[#]) &;
Lm := (Lx[#] - ⅈ Ly[#]) &;

eq1 =
 Lx[Ly[ψ[x, y, z]]] - Ly[Lx[ψ[x, y, z]]] -
   ⅈ ℏ Lz[ψ[x, y, z]] // Simplify

0

eq2 =
 Ly[Lz[ψ[x, y, z]]] - Lz[Ly[ψ[x, y, z]]] -
   ⅈ ℏ Lx[ψ[x, y, z]] // Simplify

0

eq3 =
 Lz[Lx[ψ[x, y, z]]] - Lx[Lz[ψ[x, y, z]]] -
   ⅈ ℏ Ly[ψ[x, y, z]] // Simplify

0
```

```
eq41 =
 Lz[LLx[ψ[x, y, z]]] - LLx[Lz[ψ[x, y, z]]] //
  FullSimplify;
eq42 =
 ⅈ ℏ (Lx[Ly[ψ[x, y, z]]] + Ly[Lx[ψ[x, y, z]]]) //
  Simplify; eq41 - eq42 // Simplify

0

eq51 =
 Lz[LLy[ψ[x, y, z]]] - LLy[Lz[ψ[x, y, z]]] //
  FullSimplify;
eq52 =
 -ⅈ ℏ
   (Lx[Ly[ψ[x, y, z]]] + Ly[Lx[ψ[x, y, z]]]) //
  Simplify; eq51 - eq52 // Simplify

0
eq6 =
 Lz[LLz[ψ[x, y, z]]] - LLz[Lz[ψ[x, y, z]]] //
  FullSimplify

0

eq7 = LL[Lz[ψ[x, y, z]]] - Lz[LL[ψ[x, y, z]]] //
  Simplify

0

eq8 =
 Lz[Lp[ψ[x, y, z]]] - Lp[Lz[ψ[x, y, z]]] -
   ℏ Lp[ψ[x, y, z]] // Simplify

0

eq9 =
 Lz[Lm[ψ[x, y, z]]] - Lm[Lz[ψ[x, y, z]]] +
   ℏ Lm[ψ[x, y, z]] // Simplify

0
```

```
eq10 =
 LL[ψ[x, y, z]] - Lz[Lz[ψ[x, y, z]]] +
   ℏ Lz[ψ[x, y, z]] - Lp[Lm[ψ[x, y, z]]] //
  Simplify
```

0

```
eq11 =
 LL[ψ[x, y, z]] - Lz[Lz[ψ[x, y, z]]] -
   ℏ Lz[ψ[x, y, z]] - Lm[Lp[ψ[x, y, z]]] //
  Simplify
```

0

```
eq12 =
 LL[ψ[x, y, z]] - Lz[Lz[ψ[x, y, z]]] -
   Lp[Lm[ψ[x, y, z]]]/2 - Lm[Lp[ψ[x, y, z]]]/2 //
  Simplify
```

0

```
eq13 = LL[ψ[x, y, z]] // FullSimplify
```

$$-\hbar^2 \left(-2 z \psi^{(0,0,1)}[x, y, z] + \left(x^2 + y^2\right) \psi^{(0,0,2)}[x, y, z] - 2 y \psi^{(0,1,0)}[x, y, z] - 2 y z \psi^{(0,1,1)}[x, y, z] + x^2 \psi^{(0,2,0)}[x, y, z] + z^2 \psi^{(0,2,0)}[x, y, z] - 2 x \psi^{(1,0,0)}[x, y, z] - 2 x z \psi^{(1,0,1)}[x, y, z] - 2 x y \psi^{(1,1,0)}[x, y, z] + \left(y^2 + z^2\right) \psi^{(2,0,0)}[x, y, z]\right)$$